
Examples of factoring radical ideals into prime components.
restart; 

interface(imaginaryunit=_i):
with(PolynomialIdeals):

This is the example from section 3.1 on page 112.
I := <x^2+y+z-1,x+y^2+z-1,x+y+z^2-1>;

IsRadical(I), IsPrime(I), IsZeroDimensional(I);

G := Groebner[Basis](I,plex(x,y,z));

f := factor(G[1]);

One way to try to compute the radical is to throw  the square-free part of  this 
polynomial into I
I := <I,z*(z^2+2*z-1)*(z-1)>;

IsRadical(I), IsPrime(I), IsZeroDimensional(I);

G := Groebner[Basis](I,plex(x,y,z));

f := factor(G[1]);

Now we are going to split I into three components  corresponding to the three 
factors of   and then we will have  =    .

P1 := Quotient(I,<quo(f,z,z)>);
P2 := Quotient(I,<quo(f,z-1,z)>);
P3 := Quotient(I,<quo(f,z^2+2*z-1,z)>);

We could have done this by putting the three irreducible factors of   into the basis for I 
this way
Groebner[Basis](<I,z>,plex(x,y,z));
Groebner[Basis](<I,z-1>,plex(x,y,z));
Groebner[Basis](<I,z^2+2*z-1>,plex(x,y,z));

IsPrime(P1), IsPrime(P2), IsPrime(P3);

G := factor(Groebner[Basis](P1,plex(x,y,z)));

P4, P5 := Quotient(P1,<y>), Quotient(P1,<y-1>);

IsPrime(P4), IsPrime(P5);

We are done: the prime decomposition of   is the following four components
P2, P3, P4, P5;

J := Intersect(P2,P3,P4,P5);

Groebner[Basis](J,plex(x,y,z));

The above example worked because when we computed the Groebner basis for I we found a
polynomial that factored.  This does not always happen.  Consider the following example
I := <x^2+1,y^2+1,z^2+1>;

G := Groebner[Basis](I,plex(x,y,z));

Notice that G is a Groebner basis for I in every monomial ordering since LT( ) =   in 
every monomial ordering.  But it is not prime over the field of rational numbers Q.
IsPrime(I), IsRadical(I);

We can solve this problem by inspection.  The second generator minus the first factor.
f := (x^2+1)-(y^2+1);

factor(f);

P1, P2 := Quotient(I,<x-y>), Quotient(I,<y+x>);

Again, the second generator minus the third factors into  so we repeat this 
decomposition on each component.
P11, P12, P21, P22 := Quotient(P1,<y-z>), Quotient(P1,<y+z>),
                      Quotient(P2,<y-z>), Quotient(P2,<y+z>);

map( IsPrime, [P11, P12, P21, P22] );

Intersect(P11,P12,P21,P22);

But how would we do this if we cannot "spot" a polynomial in I that factors ?
One approach is to make a linear subsitution to try to "put the ideal in general position" so 
that when we compute the lex Groebner basis with  >   >    we get the form [  , 

 ,  ] from where the problem is easily solved - we just need to look at the 
factors of  since the other polynomials are LINEAR in  and  .  This idea only works if 
the ideal is zero dimensional, i.e. the variety of the ideal has finitely many solutions.
J := < subs( z=z+5*x-3*y, Generators(I) ) >;

G := Groebner[Basis](J,plex(x,y,z));

f := factor(G[1]);

S1 := Simplify(<op(1,f),J>);

Undoing the linear change of variables we get the first prime component of I
P1 := Simplify( <subs(z=z-5*x+3*y,Generators(S1))> );

for k from 2 to 4 do
    P||k := Simplify( <subs(z=z-5*x+3*y,Generators(<op(k,f),J>))> );
od;

These components are all prime over the field of rational numbers.  A check

Simplify( Intersect(P1,P2,P3,P4) );

Simplify computes a Groebner basis for an ideal in the graded ordering by default.  The 
PrimeDecomposition command computes the prime decomposition of the radical of I.
Primes := [PrimeDecomposition(I)];

map(Simplify,Primes);


